```
> 
```

Berechne die Energie des gestreuten Photons als Funktion des WInkels des gestreuten ELektrons, theta, fuer h*nue=m*c^2.

```
> restart:
  sys1:={a*cos(theta)=mass*c^2-d*cos(Phi),a*sin(theta)=d*sin(Phi),a^2=(mass*c^2-d)^2+2*mass*c^2*(mass*c^2-d)};
```
Impulserhaltung, a=p*c,h*nue=m0*c^2,d=h*nue'

```
  sols1:=eliminate(sys1,Phi):
```
Eliminiere Phi aus den GLeichungen

```
  sols1[2];
```

$$sys1 := \left\{ a^2 = (mass\ c^2 - d)^2 + 2\ mass\ c^2\ (mass\ c^2 - d),\ a\ \sin(\theta) = d\ \sin(\Phi),\ a\ \cos(\theta) = mass\ c^2 - d\ \cos(\Phi) \right\}$$

$$\{a^2 \sin(\theta)^2 + a^2 \cos(\theta)^2 - 2\ a \cos(\theta)\ mass\ c^2 + mass^2\ c^4 - d^2,\ a^2 - 3\ mass^2\ c^4 + 4\ mass\ c^2\ d - d^2\}$$

```
> sols2:=eliminate(sols1[2],a);
```
Energieerhaltung, eliminiere a=p*c

```
  sols3:=solve(sols2[2],d);
```
Löse die verbliebene Gleichung nach d=h*nue' auf

```
  subs(sols3[2],d);
```
von den beiden Loesungen ist die zweite Nicht-Trivial, d.h. man erhaelt h*nue' als Fkt. von theta.

$$sols2 := \left[\left\{ a = \frac{2\ (mass\ c^2 - d)}{\cos(\theta)} \right\},\ \{mass\ c\ (mass\ c^2 - d)\ (3\ mass\ \cos(\theta)^2\ c^2 - 4\ mass\ c^2 + 4\ d - \cos(\theta)^2\ d)\} \right]$$

$$sols3 := \{d = mass\ c^2\},\ \left\{ d = \frac{mass\ c^2\ (3\cos(\theta)^2 - 4)}{-4 + \cos(\theta)^2} \right\}$$

$$\frac{mass\ c^2\ (3\cos(\theta)^2 - 4)}{-4 + \cos(\theta)^2}$$

```
> 
  restart:
```
Berechne die Energie des gestreuten Photons als Funktion des WInkels des gestreuten Photons, Phi, fuer h*nue=m*c^2.

```
  sys1:={a*cos(theta)=mass*c^2-d*cos(Phi),a*sin(theta)=d*sin(Phi),a^2=(mass*c^2-d)^2+2*mass*c^2*(mass*c^2-d)};
```
Impulserhaltung, a=p*c,h*nue=m0*c^2,d=h*nue'

```
  sols1:=eliminate(sys1,theta);
```
Eliminiere theta aus den GLeichungen

```
  sols1[2];
```

$$sys1 := \left\{ a^2 = (mass\ c^2 - d)^2 + 2\ mass\ c^2\ (mass\ c^2 - d),\ a\ \cos(\theta) = mass\ c^2 - d\ \cos(\Phi),\ a\ \sin(\theta) = d\ \sin(\Phi) \right\}$$

$$sols1 := \left[\left\{ \theta = \arctan\left(\frac{d\ \sin(\Phi)}{a},\ \frac{mass\ c^2 - d\ \cos(\Phi)}{a} \right) \right\},\right.$$

$$\left. \{mass^2\ c^4 - 2\ mass\ c^2\ d\ \cos(\Phi) - a^2 + d^2,\ -a^2 + 3\ mass^2\ c^4 - 4\ mass\ c^2\ d + d^2\} \right]$$

$$\{mass^2 c^4 - 2\, mass\, c^2\, d \cos(\Phi) - a^2 + d^2, -a^2 + 3\, mass^2 c^4 - 4\, mass\, c^2\, d + d^2\}$$

```
> sols2:=eliminate(sols1[2],a);
```
Energieerhaltung, eliminiere a=p*c
```
  sols3:=solve(sols2[2],d);
```
Löse die verbliebene Gleichung nach d=h*nue' auf
```
  sols4:=subs(sols3[1],d);
```
Loese nach d=h*nue' als Fkt. von theta.

$$sols2 := [\{a = \mathrm{RootOf}(\_Z^2 - 3\, mass^2 c^4 + 4\, mass\, c^2\, d - d^2)\}, \{c\, mass\, (mass\, c^2 + d \cos(\Phi) - 2\, d)\}]$$

$$sols3 := \left\{ d = -\frac{mass\, c^2}{\cos(\Phi) - 2} \right\}$$

$$sols4 := -\frac{mass\, c^2}{\cos(\Phi) - 2}$$

```
> sols5:=solve(mass*c^2*(3*cos(theta)^2-4)/(-4+cos(theta)^2)=sols4,Phi);
```
Vergleiche die erste mit der zweiten Loesung und nutze dies um Phi=Phi(theta) zu berechnen.

$$sols5 := \arccos\left(\frac{5 \cos(\theta)^2 - 4}{3 \cos(\theta)^2 - 4}\right)$$

```
> plot(sols5,theta=-Pi..Pi);
```
Plotte Phi(theta). Merke: fuer theta = Pi/2 (d=m*c^2) erhaelt man nur eine triviale Loesung...


3
2,5
2
1,5
1
0,5
0
-3
-2
-1
0
1
2
3
theta

>